


1. (1,5 puntos)

```

Sub cmd1_Click()
  Dim x As Integer, y As Integer

  pct1.Cls
  If Not IsNumeric(txt1.Text) Or Not IsNumeric(txt2.Text) Then
    pct1.Print "Los operandos no son numéricos"
  Else
    x = Val(txt1.Text)
    y = Val(txt2.Text)
    If x <= 0 Or y <= 0 Then
      pct1.Print "Los números no son positivos"
    Else
      If x Mod y = 0 Then
        pct1.Print x & " es divisible por " & y
      Else
        pct1.Print x & " no es divisible por " & y
      End If
    End If
  End If
End Sub
  
```

2. (3 puntos) Descomponiendo:



## 3. (3 puntos)

a)

```

Function Titulizar(ByVal s As String) As String
    Dim i As Integer, n As Integer
    Dim c As String ' Carácter
    Dim r As String ' Frase resultante
    Dim p As Boolean ' si ha comenzado la palabra

    p = True ' Estamos al principio
    r = ""
    s = LCase(s) ' Ponemos toda la frase en minúsculas
    n = Len(s)
    i = 1
    While i <= n
        c = Mid(s, i, 1)
        If Not esSep(c) And _
            Not esSig(c) Then
            ' Es una letra
            If p Then
                r = r & UCase(c) ' La primera letra en mayúscula
                p = False ' Ya hemos colocado la primera letra
            Else
                r = r & c ' Añadimos el carácter en minúscula
            End If
            i = i + 1
        Else ' Separador o signo
            p = True ' Ha terminado la palabra
            While esSep(c) Or esSig(c)
                ' Los signos se copian directamente, los separadores se ignoran
                If esSig(c) Then
                    r = r & c
                End If
                i = i + 1
                c = Mid(s, i, 1)
            Wend
            ' Añadimos un espacio ya sea
            ' porque uno o varios separadores se convierten en uno
            ' porque viene después de un signo de puntuación
            r = r & " "
        End If
    Wend
    Titulizar = r
End Function

```

b)

```

Sub cmd1_Click()
    Dim frase As String

    frase = txtFrase.Text ' Lee la frase sin normalizar de un campo
    txtNorm.Text = Titulizar(frase) ' Escribe normalizada en otro
End Sub

```

## 4. (2,5 puntos)

```
Sub FusionaVector(ByRef v1() As Double, ByVal n1 As Integer, _
                 ByRef v2() As Double, ByVal n2 As Integer, _
                 ByRef v3() As Double, ByRef n3 As Integer)
    Dim i1 As Integer, i2 As Integer, i3 As Integer
    i1 = 1
    i2 = 1
    n3 = n1 + n2
    For i3 = 1 To n3 Step 1 ' Copio n3 elementos en total
        If i1 > n1 Then ' Si ya he copiado todo el vector v1
            v3(i3) = v2(i2) ' Copio el elemento actual de v2
            i2 = i2 + 1 ' Incremento el índice de v2
        ElseIf i2 > n2 Then ' Lo mismo si ya he copiado todo v2
            v3(i3) = v1(i1) ' ==
            i1 = i1 + 1 ' ==
        ElseIf v1(i1) < v2(i2) Then ' Comparo los elementos
            v3(i3) = v1(i1)
            i1 = i1 + 1
        Else
            v3(i3) = v2(i2)
            i2 = i2 + 1
        End If
    Next i3
End Sub
```